MPLUS SYNTAX FOR THE COMBINED LST MODEL
     Title: Combined LST
  DATA: File is LST.dat;
  TYPE IS individual;
  Variable: NAMES ARE
  ID K11_SEX 
  Y1T1_C Y2T1_C Y3T1_C
  Y1T2_C Y2T2_C Y3T2_C
  Y1T3_C Y2T3_C Y3T3_C
  Y1T1_T Y2T1_T Y3T1_T
  Y1T2_T Y2T2_T Y3T2_T
  Y1T3_T Y2T3_T Y3T3_T;
  MISSING are all (999);
    USEvariables are
  Y1T1_C Y2T1_C Y3T1_C
  Y1T2_C Y2T2_C Y3T2_C
  Y1T3_C Y2T3_C Y3T3_C
  Y1T1_T Y2T1_T Y3T1_T
  Y1T2_T Y2T2_T Y3T2_T
  Y1T3_T Y2T3_T Y3T3_T;
  !!! Y1 FEEL SORRY FOR OTHERS
  !!! Y2 FELL SORRY FOR OTHER CHILDREN TEASED
  !!! Y3 FELL SORRY FOR OTHER CHILDREN SAD OR UPSET
  !!! C = CAREGIVERS T = TEACHERS
  Analysis:
  Estimator = MLR;
  Model:
!!!!!!!!!!!!!!!!!!!!SYMPATHY CAREGIVERS (parents) !!!!!!!!!!!!
!!!OCCASION SPECIFIC LATENT FACTORS (ZETA)
  SYP_T1 BY Y1T1_C@1;
  SYP_T1 BY Y2T1_C (G2);
  SYP_T1 BY Y3T1_C (G3);

  SYP_T2 BY Y1T2_C@1;
  SYP_T2 BY Y2T2_C (G2);
  SYP_T2 BY Y3T2_C (G3);

  SYP_T3 BY Y1T3_C@1;
  SYP_T3 BY Y2T3_C (G2);
  SYP_T3 BY Y3T3_C (G3);
!!!!!INTERCEPTS!!!!
  [Y1T1_C@0];
  [Y2T1_C] (t2);
  [Y3T1_C] (pt33); !!!!different intercept!!!

  [Y1T2_C@0];
  [Y2T2_C] (t2);
  [Y3T2_C] (t3);

  [Y1T3_C@0];
  [Y2T3_C] (t2);
  [Y3T3_C] (t3);
  !!!!!!!VARIABILITY OF STATE FACTORS!!!!!
  SYP_T1 (state1);  
  SYP_T2 (state1);
  SYP_T3 (state1);
  !!!!!!TRAIT FACTOR!!!!
  KSIP BY 		Y1T1_C@1
          		Y2T1_C (L2)
          		Y3T1_C (pL33) !!! different factor loading
         		Y1T2_C@1
          		Y2T2_C (L2)
          		Y3T2_C (L3)
          		Y1T3_C@1
          		Y2T3_C (L2)
         	 	Y3T3_C (L3);
   [KSIP]; !!!latent mean of the latent trait factor KSIP!!!
   KSIP (varP); !!!variance of the latent trait factor KSIP!!!
   !!!!! CORRELATIONS AMONG LATENT STATE AND TRAIT FACTORS FIXED TO BE ZERO!!!!!
   SYP_T1 WITH SYP_T2@0 SYP_T3@0;
   SYP_T2 WITH SYP_T3@0;
   KSIP WITH SYP_T1@0 SYP_T2@0 SYP_T3@0;

  !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!SYMPATHY TEACHER!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
	!!!!!OCCASION SPECIFIC FACTORS (ZETA)!!!!!
  SYMT_T1 BY Y1T1_T@1 ;
  SYMT_T1 BY Y2T1_T (G2);
  SYMT_T1 BY Y3T1_T (G3);

  SYMT_T2 BY Y1T2_T@1;
  SYMT_T2 BY Y2T2_T (G2);
  SYMT_T2 BY Y3T2_T (G3);

  SYMT_T3 BY Y1T3_T@1;
  SYMT_T3 BY Y2T3_T (G2);
  SYMT_T3 BY Y3T3_T (G3);

  [Y1T1_T@0];
  [Y2T1_T] (t2) ;
  [Y3T1_T] (t3);

  [Y1T2_T@0];
  [Y2T2_T] (t2);
  [Y3T2_T] (t3);

  [Y1T3_T@0];
  [Y2T3_T] (t2);
  [Y3T3_T] (t3);
!!!!!!!!VARIABILITY OF STATE FACTORS!!!!!
  SYMT_T1 (state2);
  SYMT_T2 (state2);
  SYMT_T3 (state2);
!!!!!!!!!LATENT TRAIT FACTOR KSIT!!!!!!
  KSIT BY 		Y1T1_T@1
          		Y2T1_T (L2)
          		Y3T1_T (L3)
         		Y1T2_T@1
          		Y2T2_T (L2)
          		Y3T2_T (L3)
          		Y1T3_T@1
          		Y2T3_T (L2)
          		Y3T3_T (L3);
  [KSIT]; !!!LATENT MEAN OF THE TRAIT FACTOR KSIT!!!
  KSIT (varT); !!!! VARIANCE OF THE TRAIT FACTOR KSIT!!!
!!!!!!!!!!INDEXES OF CROSS-INFORMANT CONSISTENCY AT THE STATE LEVEL!!!!!!!  
  SYP_T1 WITH SYMT_T1 (r1);
  SYP_T2 WITH SYMT_T2 (r1);
  SYP_T3 WITH SYMT_T3 (r1);
!!!!!!!!!!INDEX OF CROSS-INFORMANT CONSISTENCY AT THE TRAIT LEVEL!!!!!!!    
KSIP with KSIT;

!!!!!!! CORRELATIONS AMONG LATENT STATE AND TRAIT FACTORS FIXED TO BE ZERO!!!!!!
  SYMT_T1  WITH SYMT_T2@0  SYMT_T3@0;
  SYMT_T2 WITH SYMT_T3@0;
  KSIT WITH SYMT_T1@0 SYMT_T2@0  SYMT_T3@0;

  !CORR AMONG LATENT STATE AND TRAIT FACTORS FIXED TO BE ZERO ACROSS   INFORMANTS!!!!!!
  SYP_T1 WITH SYMT_T2@0  SYMT_T3@0;
  SYP_T2 WITH SYMT_T1@0  SYMT_T3@0;
  SYP_T3 WITH SYMT_T1@0  SYMT_T2@0;
  KSIP WITH SYMT_T1@0 SYMT_T2@0  SYMT_T3@0;
  KSIT WITH SYP_T1@0 SYP_T2@0 SYP_T3@0;
  output:
   sampstat standardized modindices (3.84) cinterval  residual tech1 tech4;
